
News 

Introducing the Power1401-3 
 

 
 
We are pleased to announce the release of the Power1401-3, our latest high 
performance data acquisition interface. It uses advanced processor technology and 
expandable memory of up to 2GB to give you the power and flexibility required for the 
most demanding applications. The significant speed increases in the Power1401-3 
make it ideal for dynamic clamp using multiple models; the table below shows the 
simultaneous input and output rates with Hodgkin-Huxley models. 

H-H Models x1 x2 x4 x8 

Power1401-3 400 kHz 300 kHz 200 kHz 110 kHz 

Power1401 mk II 130 kHz 100 kHz 60 kHz 30 kHz 

 
US Training Days: Thursday 11th and Friday 12th October 

 
Our US training days will take place at the DoubleTree by Hilton, 300 Canal Street, 
New Orleans just prior to the Society for Neuroscience meeting. These days are 
designed for both existing and prospective users and are made up of multiple threads 
to cover all aspects of sampling and analysis with Spike2 and Signal. 
 
Attendance is limited so please register early to reserve your place on the appropriate 
course. 
 

Local Training Sessions 
 
In addition to our annual UK and US training days, we are also happy to offer on-site 
local training sessions for Spike2 and Signal. These smaller training sessions can be 
tailored towards particular requirements for research groups or departments. If you are 
interested in organizing a local training seminar for your group or department, please 
contact us. 

Future meetings and events 
Cardiac & Respiratory Physiology Themed Meeting 
Hulme Hall, University of Manchester 
Manchester, UK 
September 4th - 6th 2012 
 
CED US Training Days 2012 
Doubletree by Hilton, 
New Orleans, LA, USA 
October 11th - 12th 2012 
 
Society for Neuroscience 2012 
Ernest N Morial Convention Center, 
New Orleans, LA, USA 
October 13th - 17th 2012 
 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED Downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software 
are also available. 
 
Spike2 downloads Released Signal downloads Released 
Spike2 version 7.10b 07/12 Signal version 5.07 06/12 
Spike2 version 6.16 03/11 Signal version 4.09 08/11 
Spike2 demo 07/12 Signal demo 06/12 

 
Scripts Spotlight 

I often find that when analysing data there is a need for additional processing of the 
waveform signal to enhance the features of interest. Both Spike2 and Signal can apply 
filters to data offline but it can be tedious to apply the same filter again and again to 
channels in a file or even to channels in a batch of files. Automation of these kinds of 
repetitive tasks is one of the strengths of the script language, so for this example we 
will look at creating and applying an FIR filter with the FiltCreate() and 
FiltApply() script commands. 
 
The FiltCreate()command specifies the type of filter to use, the filter specification 
and an index in the filter bank to store the filter.  
 
FiltCreate(index%, type%{, trW{, edge1{, edge2{, ...}}}}); 
 
Spike2 and Signal contain a bank of twelve filter definitions that can be viewed and 
applied from the Analysis menu FIR filters dialog. These contain default filter types 
that can be modified to suit your needs and then saved back to the filter bank, to be 
used repeatedly. The FiltCreate() function can either save the filter definition to 
an index in the filter bank using the index% argument with a value of 1-12 or create a 
temporary filter with an argument of -1 that does not overwrite any of the existing filter 
set. 
 
The FiltApply() function uses the n% argument to use an index from the filter bank 
(1-12) or a temporary filter as described above (-1). 
 
FiltApply(n%|coef[], dest%, srce%, sTime, eTime {,flags%}); 
 
The other arguments set the destination channel used to display the filtered data, the 
source waveform to filter and start and end times. 
 
The attached script code example contains a function called FiltCreateLP%(). This 

Right-click the script 

http://www.ced.co.uk/sigdemu.shtml

'$FiltCreatLP.s2s|Example function to demonstrate the use of FiltCreate() and FiltApply()

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 31/08/12

FiltCreateLP%();

Func FiltCreateLP%();
var chan%;
var edge:= 20;        'Filter cutoff
var trW:= 10;         'transition width
var ok%;
var start,fin;
var filtData%;

DlgCreate("Example Low pass filter setup");  'Start new dialog
DlgChan(1,"Channel to Low pass filter",1);
DlgReal(2,"Filter cutoff (Hz)",0.1,100000);
DlgReal(3,"Transition width",0.1,100000);
DlgButton(0,"Cancel");
DlgButton(1,"OK");
ok% := DlgShow(chan%,edge,trW);    'ok is 0 if user cancels,  variables updated if not

if ok%=1 then
    
    FiltCreate(-1,2,trW,edge);       'Create a temporary filter with the selected settings
    ChanHide(-1);                   'Hide all channels
    ChanShow(chan%);              '.. and only show selected
    XRange(0.0,MaxTime());
    CursorSet(2);                   'Set two cursors
    CursorLabel(4,1,"Start");
    CursorLabel(4,2,"End");
    Interact("Set cursors around time range to filter",1023,0);  'Ask user to select time range to filter
    CursorRenumber();    'make sure cursors are in the right order
    start:= Cursor(1);
    fin:= Cursor(2);        'Get start and end time for filtering
    filtData%:=FiltApply(-1,0,chan%,start,fin);          'Apply the filter and write filtered data to a new memory channel
    ChanShow(filtData%);               'Show the new filtered data channel
endif;
return 1;
end;

CED
File Attachment
FiltCreateLP.s2s



utilises the FiltCreate() and FiltApply() functions to generate a low-pass FIR 
filter with a user-defined cutoff. This is then applied to a selected waveform channel 
time range in a data file and the filtered result is displayed in a new memory channel.  
 

 
Q. I’m a new Spike2 user who needs to setup a simple output sequence, and my 

supervisor has suggested using the graphical editor to do this. As I’m not at all 
familiar with this I was hoping for a quick guide on how to use it. 

 
A. The main display in the graphical sequence editor window contains a 

graphical representation of the enabled DAC and digital outputs. The thicker 
blue line at the bottom of the display is the Control track. The control track can 
be clicked on and edited to set the section length, number of repeats and how 
the sequence should progress to the next section. 

 

 
 

Graphical editor 
 

Above this display are controls for stepping between and deleting individual 
pulse items and a drop-down list showing the Current Section, the first of 
which is named Initial. This section runs straight away when sampling starts 
and is generally used to set initial baseline levels for the outputs. The 
remaining 26 sections are named from Key A to Key Z and can each contain 
different time bases and output protocols that can be accessed independently 
during sampling or linked together to give fully automated control. 
 
Below and to the left of the main display is a pulse palette which contains 
seven output pulse types that can be clicked and dragged onto the output 
tracks in the main display. Digital output tracks can only accept the square 
pulse and square pulse train output types; variable height pulses, ramps, 
sinewaves and arbitrary waveforms require a DAC output. The arbitrary 
waveform option can only be added to the control track and sets aside an 
area in the 1401 memory for waveform output. Waveforms can be added from 
a data file using the Sample menu Output waveform option or from a 

Show me How! 
Click here to view a 
video tutorial on how to 
use the graphical editor 

CED
Hint
You can copy pulse output data between sequence sections in the graphical editor using the Copy.. option at the top of the dialog. This allows you to quickly setup sections that use similar output pulses

http://ced.co.uk/tutorials/spike2pulsegs


script. Arbitrary waveform settings, such as the DAC to use and number of 
cycles to output can be edited in the Play waveform tab of the sampling 
configuration. Once a pulse output has been added to an output track, the 
pulse parameters are shown and can be edited in the area below the main 
display. 

 
 The bottom row of the pulse palette area contains icons for the conditional 

statements wait and branch, a digital marker and the variable arithmetic 
option. These four icons can only be added to the control track in the main 
display and can be used to add delays, wait for times, conditions or variables 
to be met, add markers to the digital marker channel and perform basic 
variable arithmetic during a sequence.  

 
 Full details of the graphical editor functions can be found in the Spike2 on-line 

help. 
 
  
Scripts: Spike2 

Q.  I would like a way of setting an alert facility in Spike2 to indicate to my subject 
to begin a reaching task at a set time into the recording 

 
A. The attached script, TaskAlert.s2s, allows the user to set a given time into 

sampling for a message alert. The script opens a new file based on the 
current sampling configuration and creates a toolbar to start and stop 
sampling. Once the ‘alert’ time has been reached, the script changes the 
background colour, sounds a tone and also uses the Speak() function to let 
the subject know that the task should begin. You could easily modify the script 
to generate other messages if required. 

  
 

 
 
Q. Is there a section in the manual that concentrates on how to setup Signal for 

use with a Magstim unit for TMS studies? I am a new CED user and need to 
setup the system to control the Magstim using the multiple states system in 
Signal.  

 
A. TMS and Magstim control using Signal is the first in a series of Application 

Guides that will be made available for download from our website and 
contains comprehensive instructions for setting up Signal and a 1401 to 
control a Magstim device for TMS recordings. These application guides are 
laid out in the style of a lab notebook and are suitable for viewing using Adobe 
Reader or for printing as a hard copy. If you have any feedback regarding the 
style or content of these Application Guides please contact sales@ced.co.uk. 

 
 
 
Scripts: Signal 

Q. Is there any way to control older Magstim 200 series models that don’t have a 
serial line connection using Signal? 

 
A. The script, Ctrl200series.sgs, can be used to control older Magstim 200 

models that have been fitted with a parallel port modification to enable 
external control of the intensity settings. The script uses multiple states to 
control the 1401 digital outputs. It creates a suitable sampling configuration 

'$Ctrl200series.sgs|Example script to provide basic control over Magstim 200 series models that have been
'fitted with the parallel port interface option.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.


var data%;                      'View handle of data file
var sRate% := 5000;             'Sample rate of waveform channels
var fLength := 500;             'Frame length (ms);
var chargeTime := 4;            'Additional outputs frame time for charging Magstim
var sPts%;                      'Data points per frame
var sOffset% := 100;            'Pre trigger time (ms)
var sOffset;                    'Pre trigger time in seconds
var nStates% := 5;              'Number of additional states
var int%[20];                   'Array of Magstim intensity levels (one per state)
var label$[20];                 'Array of state labels  
var FrameInt[20];               'Array in intervals to next frame
var TrigDir% := 1;              'Trigger polarity

ArrConst(frameInt[],5);         'Initialise frame interval

ToolbarSet(1, "Quit");          'Toolbar for script control
ToolbarSet(3, "Settings", GetSet%);
ToolbarSet(4, "New file", New%);
ToolbarEnable(4,0);
Toolbar("TMS configuration setup",1023);


Func GetSet%();
var ok%;
var okState%;
DlgCreate("Sampling parameters");               'Set sampling parameters 
DlgInteger(1, "Sample rate",100,100000);
DlgReal(2, "Frame length(ms)",50,10000);
DlgInteger(3, "Pre-trigger time",0,500);
DlgReal(4, "Charge time",0,10);
DlgInteger(5, "Number of states",1,20);
DlgList(6, "Trigger polarity","Rising|Falling",2);
ok% := DlgShow(sRate%,fLength,sOffset%,chargeTime,nStates%,trigDir%);           'Assign return values to variables
if ok% > 0 then                                                                 'If OK pressed        
    sPts% := (fLength/1000.0)* sRate%;                                          'Get number of sample points per frame
    sOffset := sOffset% / 1000.0;                                               'Convert offset to seconds
    okState% := GetStateParams%(NStates%);                                      'Jump to function to get individual state settings
    if okState% > 0 then
        SetConfig%();                                                           'Set up configuration
        ToolbarEnable(4,1);
    endif;
endif;
return 1;
end;


'Function to get & set individual state parameters (Magstim intensity, state label and interval to next frame)
Func GetStateParams%(n%);                               
var ok%;
var i%;
DlgCreate("State setup");       'Setup Magstim settings for multiple states
DlgText("State",0,1);
DlgText("Intensity",12,1);
DlgText("Label",26,1);
DlgText("Interval(s)",36,1);
for i% := 1 to n% do
    DlgText(Str$(i%),3,i%+1);
    DlgInteger(i%,3,1,127,14,i%+1);
    DlgString(n%+i%,"",10,"",24,i%+1);
    DlgReal((n%*2)+i%,5,chargeTime + 0.1,3600,40,i%+1);
next;
ok% := DlgShow(int%[:nStates%],label$[:nStates%],frameInt[:nStates%]);
if ok% > 0 then
    if trigDir% = 1 then
        ArrAdd(int%[:nStates%],128);        
    endif;
endif;
return ok%;
end;


Func New%()
data% := FileNew(0,0);          'Open new file for sampling
if data% > 0 then
    Window(0,0,100,100);                
    FrontView(data%);
    ToolbarClear(4);
    ToolbarSet(4, "Start",Start%);

else
Message("Unable to open data file! Check that the 1401 is connected\nand switched on before running script");
Halt;endif;
return 1;
end;


Func Start%()
View(data%);
SampleStart();      'Start sampling
ToolbarClear(4);
ToolbarSet(4, "Stop",Stop%);
return 1;
end;


Func Stop%()
View(data%);
SampleStop();       'Stop sampling
ToolbarClear(4);
ToolbarSet(4, "New file", New%);
return 1
end;


Func SetConfig%()
var i%;
SampleClear();                                      'Clear existing configuration
SampleBurst(1);                                     'Burst mode sampling
SamplePorts(2);                                     'Sample 2 ports
SamplePortUnits$(0," Volt");                        
SamplePortFull(0,5);
SamplePortZero(0,0);
SamplePortName$(0,"ADC 0");
SamplePortUnits$(1," Volt");
SamplePortFull(1,5);
SamplePortZero(1,0);
SamplePortName$(1,"ADC 1");
SampleKeyMark(1);                                   'Enable keyboard marker
SampleDigMark(0);                                   'Disable digital marker
SampleAutoName$("");                                'No auto file naming
SampleAutoFile(0);              
SamplePoints(sPts%);                                'Number of data points per ADC port per frame
SampleRate(sRate%);                                 'Sample rate
SampleZeroOffset(sOffset);                          'Pre zero time
SampleTrigger(1);                                   'Turn on sweep trigger
SampleMode(3);                                      'Sample mode to fixed interval
SampleOutMode(1);                                   'Pulses outut mode
SampleDigOMask(255);                                'Enable all 8 digital outputs
SampleOutClock(0.001);                              'Set output resolution to 1ms
SampleStates(nStates%);                             'Number of states
SampleStatesMode(2);                                'Dynamic outputs state mode
SampleStatesOrder(0);                               'Numeric state ordering
SampleStatesIdle(0);
SampleStatesRepeats(1);
SampleStatesOptions(2);
SamplePause(0);
SampleWrite(1);
PulseClear(0,-1);                                                  'Clear all pulses
if trigDir% = 0 then
    PulseDataSet(0,-1,0,0);
else
    PulseDataSet(0,-1,0,128);
endif;
for i% := 1 to nStates% do                                              'Set up individual state parameters
    PulseClear(i%,-1); 
    SampleOutLength(i%,chargeTime+(fLength/1000.0));                    'Set output length to same as frame length
    SampleOutTrig(i%,chargeTime);                                       'Set outputs frame to include charge time
    PulseDataSet(i%,-1,0,int%[i%-1]);                                   'Set bit value to intensity level 
    PulseAdd(i%,-1,1,"Stimulus",chargeTime + sOffset,0.01);             'Add stimulus trigger pulse at time zero
    PulseDataSet(i%,-1,"Stimulus",128);                             'Set it to be on digital output 7 
    SampleStateLabel$(i%,label$[i%-1]);                                 'Set state label
    SampleFixedInt(i%,frameInt[i%-1]);                                  'Set interval to next frame
next;
return 1;
end;




CED
File Attachment

'$TaskAlert.s2s|Opens a new data file ready to sample controlled by a script toolbar.
'Once the file is open the user can set a time into sampling in (s) for an alert for the subject
'to begin a task. This alert uses a tone output and the Speak() command as well as a message box

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

var data%;                      'Handle of new data file (Spike2 sets it to 0)
var sTime;                      'The last time we looked at the idle routine
var ok%;
var Alert%;
var ct%;
var r,g,b;
var rsub:=0.0;
var bsub:=0.0;

HideAll();                      'Hide all unwanted bars
ToolbarVisible(1);              'Make toolbar visible always
New%();                         'Set up new sampling window
DoToolbar();                    'Do the toolbar
RestoreAll();                   'Restore all bars
Halt;

func New%()                     'New sampling window

if data%>0 then                 'If there is already a data view open...
  View(data%);                  '...close it
  FileClose();                  'No point setting data% to 0...
endif;

data%:=FileNew(0,1);            'Open a new data file for sampling
if data%<0 then Message("Unable to open new data file! Check 1401 is connected\nand switched on before running script");
    RestoreAll();                   'Restore all bars
Halt() endif;
DrawMode(-1,2);                 'Set event draw mode to lines
Window(0,0,100,100);             'Make data window in top bit of screen
XRange(0,10);

ct%:=0;
FrontView(data%);               'Bring the data view to the front
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(2,1);             'Disable "Sample stop" button
DlgCreate("Alert setup");
DlgInteger(1,"Alert at sampled time: (s)",0,10000,0,0,1);
ok% := DlgShow(Alert%);

If ok%=0 then
    Message("Alert not set");       'If cancel is pressed
endif

ToolbarText("Press SAMPLE START to commence sampling");
return 1;
end;

proc DoToolbar()
ToolbarSet(1,"&Quit",Quit%);    'Set up toolbar buttons
ToolbarSet(2,"&Sample start", Start%);
ToolbarSet(3,"&Sample stop", Stop%);
ToolbarSet(4,"&New file", New%);
ToolbarEnable(3,0);             'Disable "Sample stop" button
Toolbar("Press SAMPLE START to commence sampling", 1023);'Wait here until quit
end;

func Quit%()                    'If "Quit" is pressed
SampleStop();                   'Stop sampling
RestoreAll();
return 0;                       'leave toolbar
end;

func Start%()                   'If "Start" is pressed
ToolbarSet(0,"",Idle%);         'Call Idle%() whenever there is free time
Seconds(0);
SampleStart();                  'Start sampling
ViewColourGet(0,r,g,b);
if r=1.0 then bsub:=1.0;
else
    rsub:=1.0;
    endif;
ToolbarEnable(4,0);             'Disable "New file" button
ToolbarEnable(3,1);             'Enable "Sample stop" button
ToolbarEnable(2,0);             'Disable "Sample start" button
ToolbarEnable(1,0);             'Disable "Quit" button
ToolbarText("Press SAMPLE STOP to stop sampling");
return 1;                       'Stay with toolbar
end;

func Stop%()                    'If "Stop" is pressed
SampleStop();                   'Stop sampling
ToolbarClear(0);
if (data%>0) and                'if no data in file, stop will close it
   (ViewKind(data%) <> 0) then data% := 0 endif;
ToolbarEnable(4,1);             'Enable "New file" button
ToolbarEnable(3,0);             'Disable "Sample stop" button
ToolbarEnable(1,1);             'Enable "Quit" button
ToolbarText("Press FILE NEW to capture more data");
return 1;                       'Stay in toolbar
end;

func Idle%()                    'Idle routine runs when system has time
var eTime;
if ViewKind(data%)=0 then       'This checks that the data view exists...
   View(data%);                 '...as this gives an error if it doesn't
  if ct%=0 then
        If seconds()>= Alert% then      'Once alert time is reached
            ViewColourSet(0,rsub,0,bsub);
            DrawAll();
            Sound(60,1,1);              'Play a tone
            Speak("Time to begin task");
            if r=0.0 and g=0.0 and b=0.0 then
                ViewColourSet(0);
            else
                ViewColourSet(0,r,g,b);
           endif;
            ct%:=1;
            rsub:=0.0;
            bsub:=0.0;
    Endif   
endif

   endif;
return 1;                       'Stay in toolbar
end;

'You can find this code in the App() on-line help.
var gFloat%[20];                'global for floating window states
proc HideAll()
var i%;
gFloat%[0] := App(-2);          'number of windows
for i% := 1 to gFloat%[0] do    'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do    'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

and allows the user to specify the Magstim output intensities and interval 
between outputs for each state used. The connection between the Magstim 
parallel port interface and the 1401 digital outputs can be achieved with a 
modified cable, available from CED. 

 

Did you know...? 

The Power1401-3 can be fitted with programmable gain of x1, 2, 5 and 10 for the ADC 
inputs on the base unit and top-boxes. 

 

Recent Questions 
Q. I would like to record pulses from an external device to my Spike2 data file, 

but these pulses are only 100ns wide and are not always registered by the 
digital inputs.  

 
A. The digital inputs on the Micro1401 and Power1401 can be used to log 

external pulses of 1us or wider as events in a Spike2 data file. For pulses 
narrower than 1us, you would need to utilise a separate device or circuit 
between the pulse generator and the 1401 that has the effect of ‘stretching’ 
the pulse, allowing it to be recognised by the 1401 digital inputs. Contact us 
for details. 

 
The digital input bits 8 and 9 are routed to the Event Input front panel BNC’s 
by default, with the remaining bits 10-15 available on the rear digital input port. 
You can set Spike2 to read all of the digital inputs from the rear port in the 
Edit menu Preferences dialog. Click on the Sample tab and select the 
Event ports 0 and 1 on rear digital input connector option. 
 

 
User group  
Try the CED Forums bulletin board for software and hardware support 
 
If you have any comments about the newsletter format and content, or wish to 
unsubscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 
 
All Trademarks are acknowledged to be the Trademarks of the registered holders. 
 

Back to top 

Back to top 

http://www.ced.co.uk/phpBB3/index.php
mailto:sales@ced.co.uk
CED
Hint
We can manufacture a wide range of custom cables to connect external equipment to a 1401. Contact us for details if you require a custom connection for your hardware.

